

CLASE PSPICE

El archivo fuente (texto extensión .CIR) debe contener:

1ra línea: Título
 Última línea: .END

Se deben numerar todos los nodos (de dos o mas elementos) del circuito (números positivos) y tomar un nodo de referencia (nodo 0). Cada nodo debe tener un camino de d.c. al nodo de referencia. El circuito no debe contener lazos de fuentes de tensión ni de inductores.

Descripción de elementos:

(Si no se especifica el nodo Nb , el valor por defecto es cero)

Fuente de tensión DC:

Vxxx Na Nb DC *Valor(V)*
 (Nb nodo de referencia de fuente)

Fuente de tensión sinusoidal:

Vxxx Na Nb AC *Valor pico(V) fase (grados)*
 (la frecuencia de esta fuente se especifica en el análisis de AC)

Fuente de corriente de DC:

Ixxx Na Nb DC *Valor(A)*
 (Nb nodo de referencia de fuente)

Fuente de corriente sinusoidal:

Ixxx Na Nb AC *Valor pico(A) fase (grados)*
 (la frecuencia de esta fuente se especifica en el análisis de .AC)




Fuente de tensión lineal a tramos:

Vxxx Na Nb PWL ($T0 X0 T1 X1...$)
 ($X1, X2$, en Volts)

Fuente de corriente lineal a tramos:

Ixxx Na Nb PWL ($T0 X0 T1 X1...$)
 ($X1, X2$, en Amperes)



Fuente de tensión sinusoidal:

Vxxx Na Nb SIN ($V0 VA F TD TE$)

Representa una fuente de valor:

0 para $0 < t < TD$

$V0 + VA \text{ sen}[2\pi F(t-TD)]e^{-TE(t-TD)}$; con $V0, VA$ en volts, t en segundos.

Fuente de corriente sinusoidal:


Ixxx Na Nb SIN ($A0 AA F TD TE$)

Representa una fuente de valor:

0 para $0 < t < TD$
 $A0 + AA \text{ sen}[2\pi F(t-TD)]e^{-TE(t-TD)}$; con $A0, AA$ en volts, t en segundos.

Fuente de tensión pulsante periódica:

Vxxx Na Nb PULSE (V1 V2 TD TR TF AP PER)



Resistencias, inductores y condensadores:

Rxxx Na Nb Valor (ohms)
Lxxx Na Nb Valor (Hy) IC=x
(La condición inicial es una corriente que va de Na a Nb)
Cxxx Na Nb Valor (F) IC=x
(La condición inicial es una tensión referida a Nb)

Escribir el siguiente ejemplo (modo texto, hacer una gráfica del circuito) y guardarlo como ejemplo1.CIR

```
Ejemplo1
R1 1 2 1000
R2 2 0 1000
V1 1 0 DC 10
C1 2 0 1e-6
.END
```

Abrir PSPICE AD BASIC, →FILE, →OPEN, ejemplo1.CIR. Ejecutar SIMULATION →RUN ejemplo1, y luego VIEW→OUTPUT FILE. (Se generó Ejemplo1.OUT).

Cuando se ejecuta un archivo que no contiene ningún enunciado de control, se obtiene el análisis de DC (corrientes, tensiones y potencias) en el archivo de salida

Enunciados de control:

Describen el análisis que se realizará.

Enunciado .OP:

Agrega al archivo de salida los puntos operativos de los dispositivos no lineales, y algunos parámetros del modelo de señal evaluados en el punto de operación.

Enunciados .DC y .PRINT:

Permite incrementar el valor de una fuente independiente en un intervalo especificado.

.DC nombre de la fuente Valor inicial valor final incremento

Puede especificarse a continuación una segunda fuente, para cada valor de la cual se analiza la variación de la primer fuente.

Ejemplo: .DC V1 0 10 1 V2 0 10 1

Este enunciado no produce salida en el archivo .OUT. Para obtenerla debe usarse el siguiente enunciado:

.PRINT DC Variables de salida

Donde las variables de salida pueden especificarse como

V(N1, N2) (tensión entre nodos n1 y N2).
 I(Vxxx) (corriente por una fuente, entrante al nodo positivo).
 I(componente) (corriente por un componente, del 1er al 2do nodo).

Por ejemplo, modifíquese el ejemplo 1 de la siguiente manera, y obsérvese el archivo .OUT.

```
Ejemplo2
R1 1 2 1000
R2 2 0 1000
V1 1 0 DC 10
C1 2 0 1e-6
.DC V1 5 10 1
.PRINT DC V(2) I(V1) I(R2)
.END
```

Enunciado .PROBE:

Genera un archivo de salida .DAT, y permite la graficación de las variables.

Por ejemplo, agréguese al ejemplo 2 este comando, y córrase la simulación. Aparece una ventana gráfica. Hágase PLOT→Axis Settings→Axis variable, lo que permite seleccionar el eje x. TRACE→Add Trace permite seleccionar la curva a graficar.

Enunciado .TRAN:

Provee la respuesta temporal de un circuito. El formato es

```
.TRAN TINCR TFIN <UIC>
```

donde TINCR es el incrementode graficación, TFIN es el tiempo final de simulación. Si se omite UIC (usar condicionesd iniciales), PSPICE determina primero el punto operativo antes de iniciar el análisis transitorio. Este comando no genera salida, y debe utilizarse junto con el comando .PROBE o:

```
.PRINT TRAN Variables
```

Por ejemplo, modifíquese el ejemplo 1 de la siguiente manera y simúlese:

```
Ejemplo TRAN
R1 1 2 1000
R2 2 0 1000
V1 1 0 DC 10
C1 2 0 1e-6
.TRAN 0.0001 0.005 UIC
.PROBE
.END
```

Luego gráfiquese la tensión V(2).

Enunciado .AC:

Provee la respuesta en frecuencia de un circuito. El formato es

```
.AC TIPO NP FINICIO FFIN
```

donde TIPO se especifica LIN (análisis lineal), OCT (análisis por octavas) o DEC (análisis por décadas), NP especifica el número de puntos (total, por octavas o por décadas) y FINICIO y FFIN rerepresentan las frecuencias de inicio y fin de la simulación.

En este análisis todas las fuentes de AC operan a la misma frecuencia. Para obtener la salida se debe agregar el comando PROBE o el comando

```
.PRINT AC Vx(N1,N2) .....Ix(N1,N2)
```

donde x debe ser una de las siguientes letras:

M: Magnitud.
 P: Fase.
 R: Parte real.
 I: Parte imaginaria
 DB: $20\log_{10}$ (magnitud)

Por ejemplo, modifíquese el ejemplo 1 de la siguiente manera:

```
EjemploAC
R1 1 2 1000
R2 2 0 1000
V1 3 0 DC 10
V2 1 3 AC 1
C1 2 0 1e-6
.AC DEC 20 10 10000
.PRINT AC VM(2) VP(2)
.PROBE
.END
```

Obsérvese cómo se modifica el archivo .OUT. Luego hágase en la ventana de graficación TRACE→Add Trace y escríbase en “Trace Expression” VDB(2). Luego hágase PLOT→Add Plot to the Window, para luego hacer TRACE→Add Trace y tipear VP(2). Se obtendrá el gráfico de Bode completo. Otra forma, es escribir en “Trace Expression” (ver menú “Plot Window Templates” de TRACE→Add Trace) Bode Plot dB - separate(V(2)).

Enunciado .STEP:

Este comando no es soportado por OrCAD PSpice A/D Basics Package, pero sí en versiones anteriores de PSPICE.

Permite variar el valor de un elemento del circuito, analizando el circuito para cada valor. El formato es el siguiente

```
.STEP TIPO NOMBRE INICIO FIN INCR
```

donde TIPO especifica el tipo de variación LIN (lineal), OCT (logarítmica por octavas) o DEC (logarítmica por décadas), NOMBRE especifica el nombre del modelo de componente definido en un enunciado .MODEL (componente que se variará), INICIO y FIN representan el valor inicial y final del elemento, e INCR especifica el tamaño del incremento, para variación lineal, o el número de puntos por octavas o por décadas, para las otras variaciones.

Otra forma de especificar el comando, es usando una lista de valores discretos del componente a variar:

```
.STEP NOMBRE LIST V1 V2 v3
```

donde V1, V2, etc. Representan la lista de valores.

El formato para el enunciado .MODEL es el siguiente:

```
.MODEL NOMBRE TIPO (PARÁMETROS)
```

donde TIPO es el nombre estándar del dispositivo, (RES, CAP, IND o D para especificar resistencias, capacitores, inductores, o diodos) y PARÁMETROS es la lista de parámetros que caracterizan al dispositivo (R=xx, C=xx, L=xx, IS=xx, etc.)

El valor que se usa para la simulación, es el valor definido en el comando STEP, multiplicado por el valor que figura en el enunciado MODEL, y multiplicado por el valor que figura en la línea que define el componente.

Por ejemplo, córrase el siguiente archivo:

```
Ejemplo STEP
R1 1 2 1000
V1 1 0 DC 10
R2 2 0 Rvariable 1
.MODEL Rvariable RES (R=1)
.STEP RES Rvariable(R) LIST 500 1000
.END
```

Bibliografía adicional: Introducción a PSPICE, J. Nilsson y S. Riedel, Addison-Wesley Iberoamericana, 1994.